Generating crosstabs in
VFP
Several tools make it easy to change from normalized data to crosstabbed
data with both rows and columns determined by the data in the set.

Tamar E. Granor, Ph.D.
One of the questions people often ask is how to
convert their normalized VFP data into a crosstab,
with each data item from a specified column turning into the header for a separate column. There
are several tools for creating crosstabs from Visual
FoxPro data.
To start exploring crosstabs, we’ll start with something simpler. The sample Northwind company
has employees only in the US and the UK. Suppose you want to know how many there are in each
country. A simple query, shown in Listing 1, gives
you the answer; there are 4 UK employees and 5 US
employees.
Listing 1. It’s easy to get one record for each value of interest.
Here, each record indicates how many employees are in a
given country.
SELECT country, COUNT(*) ;
FROM Employees ;
GROUP BY Country ;
INTO CURSOR csrEmpsByCountry

But what if you want to know how many there
are in each job in each country? Again, that’s not
hard. Listing 2 gives us one record for each combination of job title and country; results are shown in
Figure 1.
Listing 2. Group on more fields to break the data down into
smaller groups.
SELECT Title, Country, COUNT(*) ;
FROM Employees ;
GROUP BY Title, Country ;
INTO CURSOR csrEmpsByCountry

But what if what you really want is one record
for each job title with a column for each country
showing how many employees in that country
have that job? That’s a crosstab. For a simple case
like this, you can get what you need with IIF(), as
in Listing 3 (included in this month’s downloads
as JobsByCountry.PRG). Figure 2 shows the results,
with one row for each job title and one column for
each country.
Listing 3. Here, combining SUM() with IIF() counts the number
of employees in each country with each job.
SELECT Title, ;
SUM(IIF(Country="US",1,0)) AS nUS,
SUM(IIF(Country="UK",1,0)) AS nUK ;
FROM Employees ;
GROUP BY Title ;
INTO CURSOR csrJobsByCountry

Figure 2. A crosstab uses data to determine what columns are
in the result.

This approach works fine when you know the
exact number of columns you want and it’s not too
many. But more often, you don’t know how many
columns will be in the result, and you may not even
know the set of values they’re drawn from. That’s
where a crosstab tool comes in handy.

VFPXTab

Figure 1. Grouping on Title and Country lets us see how
many employees have each job in each country.

There’s been a crosstab tool in the box since FoxPro 2.0, where it was called GenXTab. The version
that comes with VFP is VFPXTab.PRG and it’s
found in the VFP home folder. The system variable
_GENXTAB points to it; the Query Designer uses
_GENXTAB when you create a crosstab query.

Comments at the top of VFPXTab.PRG explain
its parameters and show how to call it, including
the structure of the data needed for it to work.
Basically, it expects to find a table or cursor with
three columns. By default, the data in the first column becomes the rows in the result, the data in the
second column becomes the columns in the result,
and the data in the third column is aggregated
(summed, by default) to form the data in the result.
For example, suppose we want the total sales
for each salesperson for each year with a row for
each salesperson and a column for each year. We
start with a query to collect the relevant data,
shown in Listing 4; Figure 3 shows partial results.
Listing 4. This query gives us one record for each employee
for each year, showing total sales.
SELECT EmployeeID, ;
YEAR(OrderDate) AS OrderYear, ;
SUM(Quantity*UnitPrice) AS OrderTotal ;
FROM Orders ;
JOIN OrderDetails ;
ON Orders.OrderID =
OrderDetails.OrderID ;
GROUP BY 1, 2 ;
INTO CURSOR csrYearlyTotals

Figure 4. This crosstab has one row per employee and one
column per year

A complete program collecting the data and
running the crosstab, as well as timing the crosstab
process, is included in this month’s downloads as
SalesPersonAnnualSales.PRG. The downloads also
include SalesPersonMonthlySales.PRG, which generates a crosstab with one row per salesperson and
a column for each month of 1997.
As the size of the data set supplied to VFPXTab
grows, generating a crosstab gets much slower. In
Listing 6, the result has one row for each date in the
data set and a column for each salesperson; Figure
5 shows partial results. This example is included in
this month’s downloads as SalesPersonDailySales.
PRG.
Listing 6. Here, the final result has one row per day with one
column per salesperson.
SELECT OrderDate, ;
EmployeeID, ;
SUM(Quantity*UnitPrice) AS OrderTotal ;
FROM Orders ;
JOIN OrderDetails ;
ON Orders.OrderID = ;
OrderDetails.OrderID ;
GROUP BY 1, 2 ;
INTO CURSOR csrDailyTotals
LOCAL oXTab
oXTab = NEWOBJECT("genxtab", _GENXTAB,'', ;
"csrXTab", .t.)
oXTab.MakeXtab()

Figure 3. The query in Listing 4 results in one record per
salesperson per year.

To get the results we want, we instantiate
the GenXTab class in VFPXTab.PRG and call its
MakeXTab method, as in Listing 5. The class takes
10 parameters that configure its behavior. (Because
I don’t recommend actually using this crosstab tool, I’m not going to go into detail about the
parameters. They’re documented in the code.) The
two parameters passed in the example indicate that
the result should be stored in a cursor (rather than
a table) named csrXTab. Figure 4 shows the result.
Listing 5. The GenXTab class in VFPXTab accepts up to 10
parameters.
oXTab = NEWOBJECT("genxtab", _GENXTAB,'', ;
"csrXTab", .t.)
oXTab.MakeXtab()

Figure 5. Partial results for a crosstab of daily sales by salesperson. There are more rows and more columns in the full
result.

In my tests, the crosstab with one row per salesperson and one column per year varies from about
1/10th of a second to about 1/100th of a second, but
FoxRockX

the daily sales example takes nearly half a second.
(Note that, in each case, I’m timing only crosstab
generation, not the query that assembles the data.)

FastXTab

Fortunately, there’s a faster way to get crosstabs.
FastXTab was created by Alexander Golovlev
specifically to address VFPXTab’s speed issues.
It’s available for download from the Universal
Thread at https://www.universalthread.com/
ShowHeaderDownloadOneItem.aspx?ID=9944.
FastXTab replaces VFPXTab’s long list of parameters
(which were a throwback to the program’s nonOOP origins) with properties. Table 1 shows the
key properties.
Table 1. FastXTab uses properties for configuration.

Property

Default value Purpose

cOutFile

“xtabquery”

Alias of the
output table or
cursor

lCursorOnly

.F.

Indicates whether
the results are
placed in a table
(.F.) or cursor (.T.)

lCloseTable

.T.

Indicates whether
the source table
should be closed
after creating the
crosstab

nRowField

1

Indicates which
column of the
source table
provides the rows

nColField

2

Indicates which
column of the
source table
provides the
columns

nDataField

3

Indicates which
column of the
source table
provides the data
to be aggregated

lTotalRows

.F.

Indicates whether
the result should
include an extra
row totaling the
data in the other
rows

lBrowseAfter .F.

Indicates whether
the result should
be opened in a
Browse window

The use of properties makes code that calls
FastXTab easier to read than code that calls
VFPXTab. Listing 7 shows the FastXTab equivalent of Listing 5; it’s based on the same query
(shown in Listing 4) and, of course, produces the
same results. The complete code, including timing test is included in this month’s downloads as
SalesPersonAnnualSalesFast.prg.
Listing 7. With FastXTab, the options you choose are easy to
understand because you specify them with properties.
oXTab = NEWOBJECT("fastxtab", "fastxtab.prg")
WITH oXTab AS FastXTab OF "fastxtab.prg"
.cOutFile = "csrXtab"
.lCursorOnly = .T.
.lCloseTable = .T.
.RunXtab()
ENDWITH

For this example, FastXTab isn’t faster than
VFPXTab. But for the daily sales example, you can
see the major improvement FastXTab offers. Using
the same original query as in Listing 6 and the call to
FastXTab shown in Listing 7, on my computer, the
crosstab is computed in about 2/100th of a second, that
is, more than 2000% faster than with VFPXTab. The
code, including timing test, is included in this month’s
downloads as S
 alesPersonDailySalesFast.prg.

FastTab 1.6

The additional speed and readability are enough to
make me recommend using FastXTab rather than
VFPXTab. But, as the infomercials say, “wait …
there’s more.”
Community member Vilhelm-Ion Praisach has
extended FastXTab considerably, both for usability
and to provide additional capabilities. It’s included
in this month’s downloads AS FastXTab16.ZIP.
Praisach’s documentation (as well as the latest version) can be found at http://praisachion.blogspot.
com/2015/02/fastxtab-version-16.html. The documentation includes lots of examples; I’ve broken
those examples out into individual PRGs, which are
included as FastXTab16Demos.ZIP in this month’s
downloads.
Among the things supported in Praisach’s
version are specifying the relevant fields by name
rather than column number; specifying multiple
data columns; specifying the function to apply
to a given data column, including with a custom
expression; and filtering the data source.
All the properties supported by the original
FastXTab remain, but there are quite a few new
properties to provide new capabilities. Table 2
shows some of them.

Table 2. FastXTab 1.6 includes many new properties to support
new capabilities.

Property

Default Purpose
value

cRowField

“”

cColField

cDataField

“”

“”

Expression from the
source table that
provides the rows

Name of the column
in the source table
that provides
the data to be
aggregated

“”

Expression from the
source table that
provides “pages”

nFunctionType

1

Indicates how to
aggregate the data;
see Table 3.

cFunctionExp

“”

Expression to
use for custom
aggregation

cCondition

“”

Filter to apply to
source data before
aggregating

cHaving

“”

Filter to apply to
source data after
aggregating

nMultiDataField

1

Indicates how many
data fields are
specified

anDataField

Table 3. FastXTab 1.6 offers six ways of aggregating the data.

Expression from
the source table
that provides the
columns

cPageField

Array of data fields
specified by name

anFunctionType

Array of aggregation
function choices

acFunctionExp

Array of custom
aggregation
expressions

Meaning

1

Sum

2

Count

3

Average

4

Min

5

Max

6

Custom
expression
specified in
cFunctionExp (or
acFunctionExp
for the relevant
column)

Listing 8. FastXTab 1.6 lets you specify the aggregation
function to apply, as well as specify expressions for rows and
columns.
SELECT EmployeeID, OrderDate, OrderID ;
FROM Orders ;
INTO CURSOR csrOrders
LOCAL oXTab AS FastXTab OF ;
"fastxtab16\fastxtab.prg"

Calling FastXTab 1.6 is the same as calling the
original FastXTab (which I’ll refer to as FastXTab
1.0), except that you need to point to the FastXTab
1.6 version of FastXTab.PRG. This month’s downloads include SalesPersonAnnualSalesFast16.PRG,
which is identical to SalesPersonAnnualSalesFast.
PRG, except for the path to the class library. In my
tests, it runs just as fast as the FastXTab 1.0 version.

nFunctionType

The code in Listing 8 (in this month’s downloads
SalesPersonAnnualSalesCountFast16.prg) shows
another cool feature of FastXTab 1.6, the ability to
use an expression to specify the rows or columns
rather than just a field name. To get the correct
results here, we need to see every order, but we want
them counted by year. So the query that gathers the
data keeps the original OrderDate column, but the
.cColField property specifies “YEAR(OrderDate),”
so that the result has one column per year. (In
fact, the data collection query is unnecessary here.
FastXTab could be run directly against the Orders
table.) Figure 6 shows the results.

Array of data fields
specified by column
position

acDataField

But FastXTab 1.6 lets you do much more.
Suppose you want to know how many sales each
salesperson had in each year. Specifying 2 for the
nFunctionType property indicates Count; Table 3
shows the options for this property.

oXTab = NEWOBJECT("fastxtab", ;
"fastxtab16\fastxtab.prg")
WITH oXTab AS FastXTab OF "fastxtab.prg"
.nFunctionType = 2 && Count
.cRowField = "EmployeeID"
.cColField = "YEAR(OrderDate)"
.cDataField = "OrderID"
.cOutFile = "csrXtab"
.lCursorOnly = .T.
.lCloseTable = .T.
.RunXtab()
ENDWITH

What if you want to know not just total sales or
the number of sales for each salesperson for each year,

oXTab = NEWOBJECT("fastxtab", ;
"fastxtab16\fastxtab.prg")
WITH oXTab AS FastXTab OF ;
"fastxtab16\fastxtab.prg"
.cRowField = 'EmployeeID'
.cColField = 'YEAR(OrderDate)'
.nMultiDataField = 3
.acDataField[1] = 'OrderTotal'
.anFunctionType[1] = 1
.acDataField[2] = 'OrderTotal'
.anFunctionType[2] = 3
.anDataField[3] = 'OrderID'
.anFunctionType[3] = 2
.cOutFile = "csrXtab"
.lCursorOnly = .T.
.lCloseTable = .F.
.RunXtab()
ENDWITH
Figure 6. Here, the nFunctionType property was set to 2 to count the number of
sales for each employee each year.

but the total sales, the average sale and the number
of sales? With FastXTab 1.6, you can specify multiple
data columns and apply a different aggregation
function to each. To specify multiple data columns,
set nMultiDataField to the number of data columns,
and then populate either anDataField or acDataField
with the list of data columns. If you want different
aggregation for different columns, fill anFunctionType
as well. In Listing 9 (included in this month’s
downloads as SalesPersonAnnualSumAvgCnt.prg),
nMultiDataField is set to 3. The first two columns
use the same field from the source, OrderTotal, but
each applies a different function. Figure 7 shows the
results. There are three columns for each year, one
showing the total, one the average, and one the count.
Listing 9. FastXTab 1.6 lets you specify multiple data columns.
SELECT EmployeeID, Orders.OrderID, ;
OrderDate, ;
SUM(Quantity*UnitPrice) AS OrderTotal ;
FROM Orders ;
JOIN OrderDetails ;
ON Orders.OrderID = ;
OrderDetails.OrderID ;
GROUP BY 1, 2, 3 ;
INTO CURSOR csrOrderTotals
LOCAL oXTab AS FastXTab OF ;
"fastxtab16\fastxtab.prg"

The previous examples showed
that you can use an expression to
specify the column field. In fact, you
can use an expression for the row
field, too and that expression can be
fairly complex (in either case). Listing 10 inverts the
problem we’ve been looking at, putting employees
in columns and time period in rows. In this case,
each row specifies a quarter, using an expression
that gives a result like 1998_Q1 (for the first quarter
of 1998). Figure 8 shows partial results. This query
is included in this month’s downloads as SalesPersonColsQuarterly.PRG.
Listing 10. The expressions used to specify row and columns
can be complex.
SELECT EmployeeID, Orders.OrderID, ;
OrderDate, ;
SUM(Quantity*UnitPrice) AS OrderTotal ;
FROM Orders ;
JOIN OrderDetails ;
ON Orders.OrderID = ;
OrderDetails.OrderID ;
GROUP BY 1, 2, 3 ;
INTO CURSOR csrOrderTotals
LOCAL oXTab AS FastXTab OF ;
"fastxtab16\fastxtab.prg"
oXTab = NEWOBJECT("fastxtab", ;
"fastxtab16\fastxtab.prg")
WITH oXTab AS FastXTab OF ;
"fastxtab16\fastxtab.prg"
.cRowField = 'PADL(YEAR(OrderDate),4) + ;

Figure 7.Using the nMultiDataField property, you can get multiple data columns for each value of the specified column field. Here,
there are three columns for each year.

oXTab = NEWOBJECT("fastxtab", ;
"fastxtab16\fastxtab.prg")
WITH oXTab AS FastXTab ;
OF "fastxtab16\fastxtab.prg"
.cPageField = 'ShipCountry'
.cRowField = 'CategoryName'
.cColField = 'YEAR(OrderDate)'
.cDataField = 'NumSold'
.cOutFile = "csrXtab"
.lCursorOnly = .T.
.lCloseTable = .F.
.RunXtab()
ENDWITH

"_Q" + PADL(QUARTER(OrderDate),1)'
.cColField = 'EmployeeID'
.nMultiDataField = 3
.acDataField[1] = 'OrderTotal'
.anFunctionType[1] = 1
.acDataField[2] = 'OrderTotal'
.anFunctionType[2] = 3
.anDataField[3] = 'OrderID'
.anFunctionType[3] = 2
.cOutFile = "csrXtab"
.lCursorOnly = .T.
.lCloseTable = .F.
.RunXtab()
ENDWITH

Figure 8. Each row here represents a quarter and each set of three columns represents an employee.

Both versions of FastXTab support “pages,”
the ability to group rows. (Imagine each “page” as
being a tab in a workbook.) With FastXTab 1.0, you
specify the column that determines “pages” with
the nPageField property. FastXTab 1.6’s cPageField
property lets you specify a field name or an
expression, just as you do for rows and columns.
Listing 11 (included in this month’s downloads as
ProductsSold.PRG) shows how many units of each
category were sold and shipped to each country
each year; Figure 9 shows partial results. The
most obvious use for data in this form is making
grouping in a report simpler.
Listing 11. Use cPageField to specify an expression to “page”
by in the crosstab.
SELECT ProductName, CategoryName, ;
OrderDate, ShipCountry, ;
SUM(Quantity) AS NumSold ;
FROM Orders ;
JOIN OrderDetails OD ;
ON Orders.OrderID = OD.OrderID ;
JOIN Products ;
ON OD.ProductID = Products.ProductID ;
JOIN Categories ;
ON Products.CategoryID = ;
Categories.CategoryID ;
GROUP BY 1, 2, 3, 4 ;
INTO CURSOR csrProductsSold
LOCAL oXTab AS FastXTab ;
OF "fastxtab16\fastxtab.prg"

Figure 9. Here, each row represents one category for one
country and each column represents a year. The data is the
number of units of that category shipped to that country in the
specified year.

When you set nFunctionType to 6 (or set
anFunctionType for a particular column to 6), you
can specify a custom aggregation calculation. In
Listing 12, we calculate the ratio of shipping cost
(Freight) to the order total for each month for
each customer. Figure 10 shows partial results;
the code is included in this month’s downloads as
FreightRatio.PRG

In Listing 13, we retrieve data on all orders, but
use only those from one year in the crosstab, which
shows totals for each salesperson for each month.
You might use this approach in a loop to generate a
crosstab for each year. Figure 11 shows the results.
The code is included in this month’s downloads as
SalesPersonMonthlyFilter.PRG.

Listing 12. You can set nFunctionType to 6 and cFunctionExp
to a custom aggregation calculation. Here, it’s the ratio of shipping cost to order total.
SELECT CustomerID, Orders.OrderID, ;
OrderDate, ;
SUM(Quantity*UnitPrice) AS OrderTotal,;
Freight ;
FROM Orders ;
JOIN OrderDetails OD ;
ON Orders.OrderID = OD.OrderID ;
GROUP BY 1, 2, 3, 5 ;
INTO CURSOR csrOrderTotals

Listing 13. The cCondition property filters data out of the cursor or table supplied to FastXTab 1.6. Here, we keep only one
year’s data.
SELECT EmployeeID, OrderDate, ;
SUM(Quantity*UnitPrice) AS OrderTotal ;
FROM Orders ;
JOIN OrderDetails ;
ON Orders.OrderID = ;
OrderDetails.OrderID ;
GROUP BY 1, 2 ;
INTO CURSOR csrMonthlyTotals

LOCAL nStart, nEnd
LOCAL oXTab AS FastXTab ;
OF "fastxtab16\fastxtab.prg"
oXTab = NEWOBJECT("fastxtab", ;
"fastxtab16\fastxtab.prg")
WITH oXTab AS FastXTab ;
OF "fastxtab16\fastxtab.prg"
.cRowField = 'CustomerID'
.cColField = 'PADL(YEAR(OrderDate),4)' + ;
' + "_" + PADL(MONTH(OrderDate),2,"0")'
.nFunctionType = 6
.cFunctionExp = ;
'SUM(Freight)/SUM(OrderTotal)'
.cOutFile = "csrXtab"
.lCursorOnly = .T.
.lCloseTable = .T.
.RunXtab()
ENDWITH

LOCAL oXTab AS FastXTab ;
OF "fastxtab16\fastxtab.prg"

FastXTab 1.6 lets you filter the data both before
and after it’s aggregated. Use cCondition to filter
before aggregating, and cHaving to filter afterward.

oXTab = NEWOBJECT("fastxtab", ;
"fastxtab16\fastxtab.prg")
WITH oXTab AS FastXTab ;
OF "fastxtab16\fastxtab.prg"
.cRowField = 'EmployeeID'
.cColField = 'MONTH(OrderDate)'
.cDataField = 'OrderTotal'
.cCondition = 'YEAR(OrderDate) = 1998'
.cOutFile = "csrXtab"
.lCursorOnly = .T.
.lCloseTable = .T.
.RunXtab()
ENDWITH

Figure 10. The data here is the ratio of freight cost to order total for a customer for a month.

Figure 11. This crosstab shows sales for each employee by month for the year specified in the cCondition property.

To demonstrate cHaving, we’ll extend the
freight ratio example. Perhaps you’re interested
in seeing only cases where customers seem to be
spending too much on freight charges. Add the line
in Listing 14 inside the WITH clause in Listing 12
to see only cases where a customer’s total monthly
freight charges were more than 5% of the total
orders. This version of the example is included in
this month’s downloads as FreightRatioFiltered.
PRG.
Listing 14. The cHaving property filters after aggregation.
Here, it keeps only data where the ratio of freight to order total
(for the month) is more than 5%.
.cHaving = ;
'SUM(Freight)/SUM(OrderTotal) >= 0.05'

Summing up

It should be obvious that I recommend FastXTab
1.6 over VFPXTab or FastXTab 1.0. Believe it or not,
FastXTab 1.6 has some additional capabilities not
discussed in this article. If you use crosstabs (or
if you now see how you can use them), I strongly
recommend spending some time not only with the
examples from this article, but with the ones that
Vilhelm-Ion Praisach provides.

In my next article, I’ll take a look at PIVOT,
SQL Server’s analogue to crosstabs.

